
Abstract 
Every observational study rests on the assumption that no important confounder has been left 
unmeasured, yet that assumption can never be verified from the data alone. Rather than hoping it 
holds, analysts can quantify how fragile or robust a finding is to such confounding. This paper 
introduces SAS programmers to a practical sensitivity analysis toolkit. Starting from a simulated 
observational cohort in which an unmeasured healthy user trait creates a falsely protective risk 
ratio, we compute the E-value, the minimum strength an unmeasured confounder would need 
with both the exposure and the outcome to fully explain the result away, and we visualize it with 
a bias plot. We then move beyond a single number to probabilistic quantitative bias analysis, 
placing distributions on the bias parameters, propagating them by Monte Carlo, and reporting a 
bias adjusted estimate with a simulation interval. Finally we calibrate the finding against 
negative control outcomes to expose residual systematic error.  



Introduction 
Randomization is what allows a clinical trial to attribute a difference in outcomes to the 
treatment rather than to the kinds of patients who received it. Observational studies have no such 
guarantee. They rest instead on an assumption, that adjustment for the measured covariates has 
removed all systematic differences between the exposure groups, and that no important 
confounder remains unmeasured. The difficulty is that this assumption can never be checked 
against the data. The unmeasured confounder is, by definition, not in the dataset. 

This paper reframes the no unmeasured confounding assumption from a yes or no question, 
which cannot be answered, into a quantitative one, which can. Instead of asking whether 
unmeasured confounding exists, we ask how strong it would have to be to change the conclusion, 
how far a plausible amount of it would move the estimate, and whether the signal exceeds the 
systematic error the design produces even when there is nothing to find. Three tools answer these 
questions in turn: the E-value, probabilistic quantitative bias analysis, and negative control 
calibration. Each is demonstrated with a simple, simulated example in SAS. 

 
Simulated Confounded Finding 
We construct a cohort in which the truth is known so that the sensitivity analyses can be judged 
against it. An unmeasured trait, which we can think of as a healthy user characteristic, makes a 
patient both more likely to be exposed and less likely to experience the outcome. The exposure 
itself has no effect on the outcome at all. Because the trait is unmeasured, an analysis that 
conditions only on the observed data will report a protective association that is entirely an 
artifact of confounding. 

 



/* Simulate a cohort in which an unmeasured healthy user trait  ​ */ 

/* creates a spurious protective association between exposure and  */ 

/* outcome. The true causal effect of exposure on outcome is zero. */ 

  

data cohort; 

   call streaminit(20260810); 

   do id = 1 to 10000; 

  ​ /* unmeasured confounder: healthy user trait */ 

  ​ healthy_user = rand('bernoulli', 0.50); 

  

  ​ /* the trait raises the chance of being exposed */ 

  ​ p_exposure = 0.30 + 0.25 * healthy_user; 

  ​ exposure   = rand('bernoulli', p_exposure); 

  

  ​ /* the trait lowers the chance of the outcome;  ​ */ 

  ​ /* exposure itself has no effect on the outcome ​ */ 

  ​ p_outcome = 0.20 - 0.08 * healthy_user; 

  ​ outcome   = rand('bernoulli', p_outcome); 

  

  ​ output; 

   end; 

   drop p_exposure p_outcome; 

run; 

  

/* Fit the model an analyst would run, without the unmeasured trait */ 

proc genmod data=cohort descending; 

   model outcome = exposure / dist=poisson link=log; 

   repeated subject=id / type=ind;   * robust sandwich SEs, Zou 2004; 

   estimate 'Exposure risk ratio' exposure 1 / exp; 

run; 

  

Output 

Analysis Of GEE Parameter Estimates 

Empirical Standard Error Estimates 

  

Parameter​ Estimate​ StdErr   95% Conf Limits      ​ Z​ Pr > |Z| 

Intercept​ -1.6094 ​ 0.0231  -1.6546   -1.5642  ​ -69.67​ <.0001 
exposure ​ -0.1961 ​ 0.0271  -0.2492   -0.1430   ​ -7.24​ <.0001 

  

Contrast Estimate Results 

Label                 ​ Estimate​ 95% Confidence Limits 

Exposure risk ratio     ​ 0.8220 ​0.7796  ​ 0.8672 

 
 
The simulated data reproduce the intended bias: exposure is associated with an about 18 percent 
lower risk of the outcome (RR 0.82, 95% CI 0.78 - 0.87), even though the exposure was 
generated with no causal effect at all.  
 



The E-Value 
The E-value, introduced by VanderWeele and Ding, is the smallest strength of association, 
expressed on the risk ratio scale, that an unmeasured confounder would need to have with both 
the exposure and the outcome, above and beyond the measured covariates, in order to fully 
account for the observed exposure outcome association. A large E-value means that only an 
implausibly strong confounder could explain the result away, which is reassuring. A small 
E-value means that even a weak confounder could do so, which is a warning. 

For an observed risk ratio at or above one, the E-value has the closed form RR plus the square 
root of RR times RR minus one. For a protective risk ratio below one, the estimate is first 
inverted. A second E-value is computed for the confidence limit closest to the null, which 
describes how much confounding would be needed to move not just the point estimate but the 
entire interval to include no effect.  

 

/* E-value for a risk ratio and its confidence interval         ​ */ 

%macro evalue(rr=, lcl=, ucl=); 

   data evalue_out; 

  ​ rr  = &rr; 

  ​ lcl = &lcl; 

  ​ ucl = &ucl; 

  

  ​ /* work on the RR >= 1 scale; invert protective estimates */ 

  ​ if rr < 1 then rr_star = 1/rr; 

            ​ else rr_star = rr; 

  ​ e_point = rr_star + sqrt(rr_star*(rr_star-1)); 

  

  ​ /* use whichever confidence limit sits closer to the null */ 

  ​ if rr < 1 then bound = ucl;   * upper limit is closer to 1; 

            ​ else bound = lcl;   * lower limit is closer to 1; 

  

  ​ if bound < 1 then bound_star = 1/bound; 

                ​ else bound_star = bound; 

  

  ​ if bound_star <= 1 then e_ci = 1;  * interval already crosses the null; 

  ​ else e_ci = bound_star + sqrt(bound_star*(bound_star-1)); 

  

  ​ format e_point e_ci 6.2; 

  ​ label e_point = 'E-value, point estimate' 

        ​ e_ci​ = 'E-value, confidence limit'; 

   run; 

  

   proc print data=evalue_out noobs label; 

  ​ var rr lcl ucl e_point e_ci; 

   run; 

%mend evalue; 

  

%evalue(rr=0.822, lcl=0.780, ucl=0.867); 



  

Output 

 ​ rr      lcl  ​ ucl​ E-value,   ​ E-value, 

                        ​ point estimate  confidence limit 

   0.822    0.780​ 0.867  ​1.73         ​ 1.57 

 

The interpretation of this E value is that an unmeasured confounder associated with both the 
exposure and the outcome by a risk ratio of only about 1.7 would be enough to produce the 
observed protective association out of nothing. Associations of that magnitude can be expected 
for the kind of healthy user characteristics that distinguish treated from untreated patients in 
routine care. The E-value does not prove that such a confounder exists, and that is its main 
limitation. Instead, it encourages the reader to pause and consider if a plausible confounder of 
that strength is likely.  

 
E-Value Bias Plot 
The E-value is a single number that summarizes the symmetric case in which the confounder is 
equally associated with exposure and with outcome. In reality those two associations can differ, 
and the full picture is a curve rather than a point. For the observed estimate, every combination 
of a confounder to exposure association and a confounder to outcome association that lies on the 
curve would be just sufficient to explain the result away. Combinations beyond the curve would 
more than explain it; combinations short of it would leave a residual association. The E-value is 
the point on this curve where the two associations are equal. 

 



/* Generate the E-value bias curve: every combination of a      ​ */ 

/* confounder-exposure and confounder-outcome risk ratio that would */ 

/* be just sufficient to explain the observed association away  ​ */ 

  

data bias_curve; 

   rr_star = 1/0.822;             ​ * invert the protective estimate; 

   do rr_eu = 1.25 to 6 by 0.02; 

  ​ rr_ud = rr_star*(rr_eu-1)/(rr_eu-rr_star); 

  ​ if 0 < rr_ud < 8 then output; 

   end; 

   keep rr_eu rr_ud; 

run; 

  

data evalue_point; 

   rr_eu = 1.73; 

   rr_ud = 1.73; 

run; 

  

proc sgplot data=bias_curve noautolegend; 

   series  x=rr_eu y=rr_ud / lineattrs=(thickness=2 color=CX2B6CB0); 

   scatter x=rr_eu y=rr_ud / data=evalue_point 

       ​markerattrs=(symbol=circlefilled color=CXC53030 size=10); 
   xaxis label='Confounder-exposure risk ratio' min=1 max=6; 

   yaxis label='Confounder-outcome risk ratio'  min=1 max=8; 

   title 'E-value bias plot for the biased exposure-outcome estimate (RR = 0.82)'; 

run; 

 



 
Figure 1. E-value bias plot. Any point on or above the curve represents an unmeasured 
confounder strong enough to fully explain the observed association, and the marked point is the 
symmetric E-value.  
 
Reading the curve demonstrates the trade off. If an unmeasured confounder were only weakly 
associated with the exposure, it would have to be very strongly associated with the outcome to 
account for the finding, and vice versa.  
 
Probabilistic Quantitative Bias Analysis 
The E-value asks how much confounding would be needed to erase the finding. Probabilistic 
quantitative bias analysis asks a different question: given what we believe about the unmeasured 
confounder, how far would it actually move the estimate? Instead of a single worst case, we 
specify probability distributions for the bias parameters, draw many scenarios by Monte Carlo, 
apply the corresponding correction in each, and summarize the distribution of bias adjusted 
estimates. 

For an unmeasured binary confounder the bias parameters are its prevalence among the exposed, 
its prevalence among the unexposed, and the strength of its association with the outcome. The 
bounding factor of VanderWeele and Arah converts these into a multiplicative bias factor. 



Because the observed estimate is protective and the confounder we posit biases it downward, the 
adjustment moves the estimate upward toward the null; the code applies the correction in the 
direction of the observed effect so that it generalizes to harmful estimates as well. We report the 
adjusted estimate first under systematic error alone and then with conventional random error 
layered on top, the latter giving a simulation interval that reflects total uncertainty. 

 

/* Probabilistic quantitative bias analysis for an unmeasured    ​*/ 
/* binary confounder, using the VanderWeele-Arah bounding factor ​*/ 
  

%macro pqba(n=200000, seed=20260810); 

   data pqba; 

  ​ call streaminit(&seed); 

  ​ obs_rr = 0.822; 

  ​ lcl    = 0.780; 

  ​ ucl    = 0.867; 

  ​ se_log = (log(ucl) - log(lcl)) / (2*1.96); 

  

  ​ do draw = 1 to &n; 

     ​ /* prevalence of the confounder among exposed and unexposed */ 

     ​ p1 = rand('beta', 28, 12); ​ * mean about 0.70; 

     ​ p0 = rand('beta', 10, 30); ​ * mean about 0.25; 

  

     ​ /* strength of the confounder-outcome association */ 

     ​ rr_ud = rand('lognormal')*0.12 + log(2.5);  * on log scale; 

     ​ rr_ud = exp(rr_ud);                      ​ * median about 2.5; 

  

     ​ /* VanderWeele-Arah bounding factor */ 

     ​ bf = (p1*(rr_ud-1)+1) / (p0*(rr_ud-1)+1); 

  

     ​ /* systematic-error-only adjustment */ 

     ​ adj_rr_point = obs_rr * bf; 

  

     ​ /* add conventional random error for a simulation interval */ 

     ​ adj_log_rr  = log(obs_rr) + log(bf) + rand('normal')*se_log; 

     ​ adj_rr_full = exp(adj_log_rr); 

  

     ​ output; 

  ​ end; 

   run; 

  

   proc means data=pqba n median p2.5 p97.5 maxdec=2; 

  ​ var adj_rr_point adj_rr_full; 

   run; 

%mend pqba; 

  

%pqba; 

  



Output 

Variable      ​N      ​Median  ​ 2.5th Pctl ​ 97.5th Pctl 

adj_rr_point​ 200000    ​ 1.22       ​ 1.00        ​ 1.51 

adj_rr_full ​ 200000    ​ 1.22       ​ 1.00        ​ 1.52 

 

 

Figure 2. Distribution of bias-adjusted risk ratios across 200,000 Monte Carlo draws. The 
median adjusted estimate, about RR = 1.21, sits on the “harmful” side of the null.  

Under plausible beliefs about the unmeasured confounder, the corrected estimate does not merely 
move toward the null, it crosses it. The median bias adjusted risk ratio is about 1.21 (95% CI: 
1.00 - 1.52). A result that looked like a significant 18 percent risk reduction is, once a modest 
and entirely plausible confounder is accounted for, arguably a modest risk increase. 
 
Negative Control Calibration 
A negative control outcome is an outcome that the exposure could not plausibly cause, so its true 
risk ratio is one. If a study design and its analysis were free of systematic error, the estimated 
associations for a set of such controls would scatter around one with only random variation. In 
practice they rarely do. The spread and the center of the negative control estimates map the 
residual systematic error that the design produces even when there is nothing to find.  



 
 

/* Empirical calibration against a panel of negative control    ​ */ 

/* outcomes. Each control shares the exposure and covariate set ​ */ 

/* but has no plausible causal pathway from the exposure.       ​ */ 

  

%macro nc_model(outcome=); 

   proc genmod data=cohort_nc descending; 

  ​ model &outcome = exposure / dist=poisson link=log; 

  ​ repeated subject=id / type=ind; 

  ​ ods output GEEEmpPEst=est_&outcome; 

   run; 

%mend nc_model; 

  

/* &nc_list holds nc1 - nc30, the thirty negative control outcomes */ 

%macro run_all_nc(nc_list=); 

   %local i outc; 

   %do i = 1 %to %sysfunc(countw(&nc_list)); 

  ​ %let outc = %scan(&nc_list, &i); 

  ​ %nc_model(outcome=&outc); 

   %end; 

%mend run_all_nc; 

  

%run_all_nc(nc_list=nc1-nc30); 

  

data nc_estimates; 

   set est_: indsname=src; 

   where parameter = 'exposure'; 

   log_rr = estimate; 

   se     = stderr; 

run; 

  

/* Fit the empirical null: a normal distribution over the negative  */ 

/* control log(RR) values                                      ​ */ 

proc means data=nc_estimates noprint; 

   var log_rr; 

   output out=null_fit mean=mu std=tau; 

run; 

  

data calibrated; 

   set null_fit; 

   study_log_rr = log(0.822); 

   study_se ​ = 0.0271; 

  

   z_naive = study_log_rr / study_se; 

   p_naive = 2*(1 - probnorm(abs(z_naive))); 

  

   z_calibrated = (study_log_rr - mu) / tau; 

   p_calibrated = 2*(1 - probnorm(abs(z_calibrated))); 

  

   format p_naive p_calibrated pvalue6.4; 



run; 

  

proc print data=calibrated noobs; 

   var mu tau p_naive p_calibrated; 

run; 

  

Output 

 ​ mu       tau  ​p_naive​p_calibrated 
  -0.150     0.300 ​ <.0001    ​ 0.8781 

  

 

Figure 3. Negative control funnel plot. The 30 negative control outcomes (grey) are centered 
away from the true null, showing that the design carries systematic error. Once the study 
estimate (red) is calibrated, the p-value rises from <0.001 to about 0.88.  

 
The naive analysis reports a highly significant protective association, with a p-value below 
0.001. Once that same estimate is judged against the empirical null built from thirty outcomes 
the exposure cannot plausibly affect, the calibrated p-value rises to about 0.88. The signal is not 
distinguishable from the systematic error the design produces on its own. 
 



Choosing and Reporting a Method 
The three tools are complementary rather than competing, and they answer different questions. 
The E-value is the natural default because it requires nothing beyond the estimate and its 
interval, is quick to compute, and gives a single comparable number that belongs in every 
observational paper. Its limitation is that it describes a worst case and does not draw on any 
external knowledge of how strong confounding actually tends to be. Probabilistic bias analysis is 
the right next step when such knowledge exists, because it translates beliefs about the 
confounder into a distribution of adjusted estimates rather than a single bound. Negative control 
calibration is the most powerful when the data permit assembling a credible set of null outcomes 
or exposures, because it measures systematic error empirically rather than positing it. 

 

Sensitivity Analysis Sample Result What It Tells You 

Biased estimate RR 0.82 (0.78 - 0.87) Significant protective 
association that is in truth null 

E-Value 1.73 estimate, 1.57 limit Only a modest confounder is 
needed to explain the bias 

Probabilistic bias analysis Adjusted RR 1.21 (1.00 - 1.52) Plausible confounding moves 
the estimate across the null 

Negative control calibration P = 0.88 versus p < 0.001 The signal is within the 
design’s own systematic error 

 
 
Limitations 
These methods address unmeasured confounding specifically, and they carry assumptions of 
their own. The E-value and the bounding factor summarize the confounder on the risk ratio scale 
and describe a maximum rather than an exact bias, so they can be conservative, and they require 
care in the presence of strong effect modification. The conclusions of a probabilistic bias analysis 
are only as credible as the distributions placed on the bias parameters; informed priors grounded 
in substantive knowledge are essential, because arbitrary inputs produce arbitrary outputs. 
Negative control calibration depends on the controls genuinely being null and sharing the 
confounding structure of the target question, so the selection of controls is itself a substantive 
task. The approximate conversions that bring odds ratios and hazard ratios onto the risk ratio 
scale assume a common outcome and should be used cautiously when the outcome is rare. 
Finally, none of these tools addresses selection bias or measurement error, each of which has its 
own quantitative bias analysis methods. 

 
 



 
Conclusion 
An observational estimate is incomplete until it is paired with a statement of how sensitive it is to 
the confounding that adjustment cannot reach. The no unmeasured confounding assumption 
cannot be verified, but its consequences can be quantified, and doing so is neither difficult nor 
time consuming in SAS. Across three different correction methods, a confounded result was 
exposed and each method told the same story from a different angle. Adopting even one of them 
as a routine part of every real world evidence analysis would make the resulting conclusions 
more honest and more useful. 

